

HP-41C Toggle Control Routines by Ron Knapp
 PPC Calculator Journal V9 N8 Pg 22 Dec 1982
 Uses HP82160A HP-IL & HP82163A Video Intfc.

```

01 LBL "Z"      ;Richard Nelson's Routine
02 FS? 05      ;If flag 5 set, cursor OFF
03 "\1B<"      ;1B60 hex = Cursor OFF
04 FS?C 05     ;If flag 5 set, clear it,
05 GTO 00      ;then go to output
06 SF 05       ;Flag 5 is clear; set it
07 "\1B>"      ;1B62 hex = Cursor ON
08 LBL 00      ;Output to Video Interface
09 XROM 28,35  ;OUTA - send escape sequence
10 RTN         ;23 bytes, .244 relative time
11 LBL "Y"     ;Ron Knapp's First Routine
12 FC?C 05     ;If flag 5 set, clear it
13 SF 05       ;If flag 5 clear, set it
14 FC? 05      ;Is flag 5 clear now?
15 "\1B<"      ;1B60 hex = Cursor OFF
16 FS? 05      ;Is flag 5 set now?
17 "\1B>"      ;1B62 hex = Cursor ON
18 XROM 28,35  ;OUTA - send escape sequence
19 RTN         ;22 bytes, .250 relative time
20 LBL "X"     ;Ron Knapp's Routine A
21 "\1B<"      ;1B60 hex = Cursor OFF
22 FS?C 05     ;If flag 5 set, clear it and
23 GTO 01      ;output Cursor OFF
24 SF 05       ;Flag 5 is clear, set it
25 "\1B>"      ;1B62 hex = Cursor ON
26 LBL 01      ;Output to Video Interface
27 XROM 28,35  ;OUTA - send escape sequence
28 RTN         ;21 bytes, .232 relative time
29 LBL "W"     ;Ron Knapp's Routine B
30 "\1B<"      ;1B60 hex = Cursor OFF
31 FS? 05      ;Is flag 5 set?
32 "\1B>"      ;1B62 hex = Cursor ON
33 FC?C 05     ;If flag 5 set, clear it
34 SF 05       ;If flag 5 clear, set it
35 XROM 28,35  ;OUTA - send escape sequence
36 RTN         ;20 bytes, .238 relative time
37 LBL "V"     ;Ron Knapp's Routine C
38 FS?C 05     ;If flag 5 set, clear it
39 GTO 02      ;Jump to Output Cursor ON
40 SF 05       ;Flag 5 is clear, set it
41 "\1B<"      ;1B60 hex = Cursor OFF
42 GTO 03      ;Jump to Output to Video
43 LBL 02      ;Output Cursor ON
44 "\1B>"      ;1B62 hex = Cursor ON
45 LBL 03      ;Output to Video Interface
46 XROM 28,35  ;OUTA - send escape sequence
47 RTN         ;24 bytes, .228 relative time
48 LBL "U"     ;Ron Knapp's Routine D
49 FS?C 05     ;If flag 5 set, clear it
50 GTO 04      ;Jump to Output Cursor ON
51 SF 05       ;Flag 5 is clear, set it
52 "\1B<"      ;1B60 hex = Cursor OFF
53 XROM 28,35  ;OUTA - send escape sequence
54 RTN         ;End of Cursor OFF
55 LBL 04      ;Output Cursor ON
56 "\1B>"      ;1B62 hex = Cursor ON
57 XROM 28,35  ;OUTA - send escape sequence
58 RTN         ;24 bytes, .209 relative time
59 END         ;137 BYTES total (with END)

```

```

0089C000F2005AAC05F21B3CAA05B100A805F21B3E01A72385
C000F20059AB05A805AD05F21B3CAC05F21B3EA72385C000F2
0058F21B3CAA05B200A805F21B3E02A72385C000F20057F21B
3CAC05F21B3EAB05A805A72385C000F20056AA05B300A805F2
1B3CB40003F21B3E04A72385C000F20055AA05B500A805F21B
3CA7238505F21B3EA72385C0000DAD

```

HP-41C Control Toggle Routines by Ron Knapp PPC V9 N8 P22 Dec 1982

Program Registers Needed: 20


Row 1 (1 - 5)




Row 2 (5 - 11)



Row 3 (11 - 16)



Row 4 (17 - 21)



Row 5 (21 - 27)



Row 6 (28 - 32)



Row 7 (32 - 37)



Row 8 (38 - 44)



Row 9 (44 - 49)



Row 10 (50 - 56)



Row 11 (56 - 59)

